
· Open SQL statements are a subset of Standard SQL that is fully integrated in the ABAP language. (SQL stands for standard query language.)

· Open SQL statements allow you to access the database in a uniform way from your programs, regardless of the database system installed. The database interface converts Open SQL statements to database-specific SQL statements. 

· You can search for database tables via the Application Hierarchy or the Repository Information System.
· If you have the name of a program that accesses the database table you are searching for:

· Debugger:
Start this program in debugging mode and set a breakpoint at the SELECT statement.

· ABAP Editor:
Search for the SELECT statement in the source text.

· If you know a program that contains a screen with input fields connected to the table you are looking for, choose F1 ( Technical info. Double-click the technical name of the screen field to go to the ABAP Dictionary. This is often a structure field. Double-click the data element and then use the where-used list to search for transparent tables with the field type

· To program database read access, use the Open SQL statement SELECT.

· The SELECT statement contains a series of clauses, each of which has a different task:

· The SELECT clause describes, among other things, whether the result of the selection will be several lines or a single data record and which fields of the table are to be read. 

· The FROM clause names the source (database table or view) from which the data is to be selected.  

· The INTO clause determines the internal data objects into which the selected data is to be placed. 

· The WHERE clause specifies the conditions that the selection results must fulfill. It thus determines the lines to be selected from the table.

· For information about other clauses, refer to the keyword documentation for SELECT.

· The SELECT SINGLE* statement allows you to read a single record from the database table. To ensure that you read a unique entry, all of the key fields must be filled by the WHERE clause. The asterisk * after SINGLE tells the database interface to read all columns in that line of the database table. If you want only a specific selection of columns, you can list the required fields instead.

· In the INTO clause, enter the destination where the database interface is to copy the data. The target area should be structured left justified just like the required columns of the database table.

· If you use the CORRESPONDING FIELDS OF addition in the INTO clause, you can fill the target area component by component. The system fills only those components that have identical names to columns in the database table. Note: If you do not use this addition, the system fills the target area left justified, irrespective of its structure.

· If the system finds a table entry matching your conditions, SY-SUBRC has the value zero (0).

· The SINGLE addition tells the database that only one line needs to be read. The database can then terminate the search as soon as it has found that line. Therefore, SELECT SINGLE produces better performance for single-record access than a SELECT loop if you supply values for all key fields

· If you do not use the addition SINGLE with the SELECT statement, the system reads multiple records from the database. The field list determines the columns whose data is to be read from the database.   

· The number of requested lines to be read can be restricted using the WHERE clause.

· In the WHERE clause you may enter only the field names of the database table. The name of the database table you want to access is found in the FROM clause.

· Multiple logical conditions can be added to the WHERE clause using AND, NOT or OR statements. 

· The database delivers data to the database interface in packages. The ABAP runtime system copies the data records to the target area line by line using a loop. It also enables sequential processing of all the statements between SELECT and ENDSELECT.

· After the ENDSELECT statement, you can check the return code for the SELECT loop. sy-subrc will have the value zero (0) if the database interface has found at least one record.

· After the ENDSELECT statement, sy-dbcnt contains the total number of lines read.

· The addition INTO TABLE itab causes the ABAP runtime system to copy the contents of the database interface directly to the internal table itab. This is called an array fetch. 

· Since the array fetch is not executed as a loop, do not program any ENDSELECT statement.

· If you want to add lines to the end of an internal table that is already filled, instead of overwriting it, use the APPENDING TABLE itab addition.
· sy-subrc has the value 0 if the system was able to read at least one record.

· The program must contain a data object with a suitable type for each column that is required from a database table. For program maintenance reasons, you must use the corresponding Dictionary objects to assign types to the data objects. The INTO clause specifies the data object into which you want to place the data from the database table. You can use the INTO clause in two different ways:

· Using a flat structure – You define a structure in your program that has the same fields in the same sequence as the field list in the SELECT clause. You can enter the name of the structure in the INTO clause. The contents are copied left justified. The field names of the structure are disregarded here.

· Using individual data objects –You can enter a set of data objects in the INTO clause. For example:
DATA:   gd_carrid TYPE sflight-carrid,
        gd_connid TYPE sflight-connid,
        gd_fldate  TYPE sflight-fldate,
        gd_seatsmax TYPE sflight-seatsmax,
        gd_seatsocc TYPE sflight-seatsocc.
START-OF-SELECTION.
SELECT carrid connid fldate seatsmax seatsocc
 FROM sflight
 INTO (gd_carrid, gd_connid, gd_fldate, gd_seatsmax, gd_seatsocc)
 WHERE ...

· If you use the INTO CORRESPONDING FIELDS clause, the data is placed in the fields with the same name in the target structure. 

· Advantages of this construction:

· The target structure does not have to be structured left justified in the identical way as the field list. 

· This construction is easy to maintain, because extending the field list does not require other changes to the program, as long as there is a field in the target structure that has the same name and type. 

· To place data in internal table columns of the same name using an array fetch, use the INTO CORRESPONDING FIELDS OF TABLE itab statement.

· With client-specific database tables, the first keyword must always be MANDT or client (data type CLNT).

· If you select data from client-specific tables, the system always reads the data records of the logon client, unless otherwise specified. (The ABAP runtime system automatically converts the SELECT statement.)

· If you wish to explicitly include the client field in the WHERE clause, you must set the CLIENT SPECIFIED addition after the FROM clause.

· Since this special case is not relevant for this course, the client field is generally omitted from the graphics.

· When you go to the definition of a database table in the ABAP Dictionary, you will see information on all the technical attributes of the database table. 

· The following information is useful for improving the performance of database accesses:

· Key fields – If the lines requested from the database are retrieved according to key fields, the Database Optimizer can perform access using a primary index.

· Secondary index – If the lines requested from the database are retrieved according to fields, the Database Optimizer can perform access using a secondary index. Secondary indexes are displayed in a dialog box whenever you select Indexes. You choose an index from the dialog box by double-clicking it. The system then displays a screen with additional information about that index. 

· You should program an authorization check before accessing the database. The AUTHORITY-CHECK statement first checks whether the user has the authorization containing all the required values. After the authorization check is completed, check the return code. If it is 0, the user has the required authorization and the program can be continued. If the value is not 0, the user does not have the required authorization and you should issue an appropriate message. 

· Later in this course, you will learn how to make fields on the selection screen ready for input again if you perform the authorization check right after the selection screen and how to send a message if the user does not have the required authorization

· All data in the SAP system must be protected from unauthorized access by users who do not explicitly have permission to access it.

· The system administrator assigns user authorizations when maintaining user master data. During this process, you should determine exactly which data users are allowed to access and what kind of access should be allowed. For example, you might want to allow users to display data for all airlines, but allow them to change data for only certain ones. In this case, the system must look for a combination of the activity (ACTVT) and airline carrier (CARRID) fields each time it performs an authorization check. Both fields must be filled with values during authorization creation as well (in this example, activity Change and airline carrier LH or activity Display and airline carrier *). You can ensure this by creating an authorization object composed of the ACTVT and CARRID fields that must be accessed both during the authorization assignment process and whenever your program performs an authorization check.

· Authorization objects simply define the combination of fields that need to be addressed simultaneously and serve as templates for both authorizations and authorization checks. They are organized into object classes to make it easier to find and manage them. One or more object classes may be assigned to each application.

· When you check authorizations in programs, you specify the object and values the user needs in a particular authorization to be able to access the object in question.

· In the above example, there is a check to determine whether the user has authorization for the object S_CARRID, which has the value 'LH' in the CARRID field and the value '02' (change) in the ACTVT field. The abbreviations for the possible activities are documented in the table TACT and also in the appropriate authorization objects. 

· The AUTHORITY-CHECK statement executes the authorization check for the user master record and also issues the return code. When evaluating the return code, you can specify the consequences of lacking authorization (for example, program terminates or skips some input lines).

· You must specify all fields of the object in an AUTHORITY-CHECK statement; otherwise you receive a return code not equal to zero. If you do not want to carry out a check for a particular field, enter DUMMY after the authorization field.
Example: When a change transaction is called, it is a good idea to check whether the user may change the entries for a particular airline: 

AUTHORITY-CHECK OBJECT 'S_CARRID'

ID 'CARRID' DUMMY

ID 'ACTVT' FIELD '02'.

· The most important return codes for AUTHORITY-CHECK are:

· 0:
The user has authorization containing the required values.
· 4:
The user does not have the required authorization.

· 8:
The check was not successful because not all fields of the authorization object were specified.

· The keyword documentation for AUTHORITY-CHECK contains a complete list of The check was not successful because not all fields of the authorization object were specifiedreturn codes.

· You can access several database tables using one database query, if the databse tables are logically related. To do so, you must specify the following information:

· Which database tables should be accessed?

· What should the link condition look like? In the graphic, columns from both database tables are linked. If all the field values of the linked columns match, a record is placed in a common line in the results table.

· Which columns are to be read? If a column appears in more than one database table, you must specify the table from which this column is to be read.

· You can define the link conditions either statically or dynamically.

· You define a static link or view in the ABAP Dictionary. There are several kinds of views. You can find information about views in the SAP Library under Basis Components ( ABAP Workbench ( BC ABAP Dictionary ( Views. 
· You implement a dynamic link or an ABAP join using ABAP statements. The system generates an appropriate database query in the database interface at run time. For more information, see the keyword documentation for the SELECT statement.

· If reusable components that encapsulate complex data retrieval statements are available, you should use them. There are four techniques for reusing data retrieval components:.

· Calling methods of global classes

· Calling methods of business objects

· Calling function modules

· Including logical databases – Logical databases are data retrieval programs that return logically-connected data in a hierarchical sequence. 

· You can find information on these techniques in the appropriate units.

· For detailed information on how to include logical databases, refer to the SAP Library under:
Basis Components  ABAP Programming and Runtime Environment  ABAP Database Accesses  Logical Databases.

· In addition to the SELECT statement, Open SQL contains the UPDATE, INSERT, DELETE, and MODIFY statements.

· Do not use these data base accesses, which initiate changes in the database, unless you understand the SAP transaction concept.

· All Open SQL statements support the return code sy-subrc. If a database access was successful, the return code in the system field sy-subrc is 0; if it was not successful, the return code is not 0.

· For more information about Open SQL statements, refer to the respective keyword documentation.

 Data Retrieval Exercises

	Unit: Data Retrieval

Topic: Reading Several Records Using a Loop

	At the conclusion of these exercises, you will be able to:

· Program loop database accesses

· Limit the quantity of selected data in accordance with an input parameter

	Enhance your ABAP program to issue flight schedules so that only the schedules of the airline carriers selected by the user are displayed.

The list should also include the percentage occupancy of each flight scheduled.

	Template:
ZBC400_##_GS_LIST or


SAPBC400WBS_GS_LIST
Program:
ZBC400_##_SELECT_SFLIGHT
Model solution:
SAPBC400DDS_SELECT_SFLIGHT
## stands for the two-digit group number.

	
	


1-1
Copy your executable program ZBC400_##_GS_LIST
(solution for exercise 2 in unit 3, Introduction to the ABAP Workbench) or the template  SAPBC400WBS_GS_LIST to the new name ZBC400_##_SELECT_SFLIGHT.

1-2
Change the work area type name (name suggestion: wa_flight). Now use the global structure SBC400FOCC.
Find out information on the components (fields) of this structure.

1-3
Define an input parameter for the airline abbreviation.

1-4
Now select only the flights from the database table SFLIGHT that belong to the selected airline.
Select only those fields that have a target field in the work area.

1-5
Within the loop statement, calculate the percentage occupancy for each flight. Assign the result to the PERCENTAGE field in your work area.

1-6
Extend your list output to include the percentage occupancy.

1-7
What is the first key field in the transparent table SFLIGHT?
What is the first key field in your WHERE clause?
Does the database interface still use the primary index?

Exercises

	Unit:
Data Retrieval

Topic:
Filling and Sorting an Internal Table

	At the conclusion of these exercises, you will be able to:

· Fill an internal table with data by appending data records

· Sort the content of an internal table

	Enhance your ABAP program for the output of flight departure times so that the flights appear in the list sorted by percentage occupancy.

For this purpose, append the read data records for the selected airline carrier to an internal table.

	Template:
ZBC400_##_SELECT_SFLIGHT or


SAPBC400DDS_SELECT_SFLIGHT
Program:
ZBC400_##_SELECT_SFLIGHT_ITAB
Model solution:
SAPBC400DDS_SELECT_SFLIGHT_TAB
## stands for the two-digit group number.


2-1
Copy your executable program ZBC400_##_SELECT_SFLIGHT
(solution for exercise 1 in this unit) or the template SAPBC400DDS_SELECT_SFLIGHT to the new name ZBC400_##_SELECT_SFLIGHT_ITAB.

2-2
Define an internal table (name proposal: it_flight) using a global table type as a basis that has the global structure SBC400FOCC as line type.

	Use the Where-used list in the ABAP Dictionary to find a suitable table type. 


2-3
Fill this internal table line by line by using an INSERT statement in the SELECT loop. (Remove the statements for list output. You can use the source text again later.)

2-4
Sort the internal table by percentage occupancy and display the contents of the internal table in a loop on a list.

Exercises

	Unit:
Data Retrieval

Topic:
Authorization Check

	At the conclusion of these exercises, you will be able to:

· Implement authorization checks

· Set up the program flow in a variable manner depending on the outcome of the authorization check

	Enhance your ABAP program so that the flights can be read and displayed only if the user has display authorization for the selected airline.

	Templates:
ZBC400_##_SELECT_SFLIGHT and


ZBC400_##_SELECT_SFLIGHT_ITAB


SAPBC400DDS_SELECT_SFLIGHT and


SAPBC400DDS_SELECT_SFLIGHT_TAB
Programs:
ZBC400_##_AUTHORITY_CHECK and


ZBC400_##_AUTHORITY_CHECK_2
Model solutions:
SAPBC400DDS_AUTHORITY_CHECK and



SAPBC400DDS_AUTHORITY_CHECK_2
## stands for the two-digit group number.


3-1
Copy your executable programs
ZBC400_##_SELECT_SFLIGHT (solution for exercise 1 in this unit) and ZBC400_##_SELECT_SFLIGHT_ITAB (solution for exercise 2 in this unit) or the templates SAPBC400DDS_SELECT_SFLIGHT and SAPBC400DDS_SELECT_SFLIGHT_TAB to the new names ZBC400_##_AUTHORITY_CHECK and ZBC400_##_AUTHORITY_CHECK_2.

3-2
Program an authorization check for the authorization object S_CARRID. Check the combination of selected airline and activity Display.

	[image: image38.wmf]
	
Use the respective Pattern statement in the ABAP Editor.

Search for the correct abbreviation of the activity to be checked. 


3-3
Define a constant for the abbreviation of the activity “Display” (name proposal actvt_display). Use ACTIV_AUTH for typing the data element and use this constant during the authorization check.

3-4
Execute the database access only if the user has display authorization for the selected airline. If this is not the case, display an appropriate error message on the list.  

3-5
Execute your programs; select the airline abbreviations ."AA" and "UA".

Data Retrival Solutions

	Unit: Data Retrieval

Topic: Reading Several Records Using a Loop


1-2
The global structure SBC400FOCC has the following components:
CARRID with type S_CARR_ID
CONNID with type S_CONN_ID
FLDATE with type S_DATE
SEATSMAX with type S_SEATSMAX
SEATSOCC with type S_SEATSOCC
PERCENTAGE with type S_FLGHTOCC.

1-7
The MANDT field in SFLIGHT is marked as a key field. The WHERE condition does not need this field. Despite this fact, the database interface can use the primary index in this case for selecting the data records.
Reason:
When the ABAP runtime system accesses a client-specific table, it assumes that data should be taken into account only if it belongs to the client under which the user is logged on. Thus, by default, the database interface enhances the WHERE condition with a WHERE mandt = sy-mandt condition. 

1
Model solution:


REPORT  sapbc400dds_select_sflight    .

DATA: wa_flight TYPE sbc400focc,
PARAMETERS: pa_car TYPE s_carr_id.


* Select all datasets from database table SFLIGHT
* corresponding to carrier PA_CAR:
SELECT carrid connid fldate seatsmax seatsocc
       FROM sflight
       INTO CORRESPONDING FIELDS OF wa_flight
       WHERE carrid = pa_car.

* Calculate occupation of each flight:
  wa_flight-percentage =
         100 * wa_flight-seatsocc / wa_flight-seatsmax.

* Create List:
    WRITE: / wa_flight-carrid,
             wa_flight-connid,
             wa_flight-fldate,
             wa_flight-seatsmax,
             wa_flight-seatsocc,
             wa_flight-percentage,
             '%'.

  ENDSELECT.
Solutions

	Unit:
Data Retrieval

Topic:
Filling and sorting an internal table


2
Model solution:

REPORT sapbc400dds_select_sflight_tab.

DATA: wa_flight TYPE sbc400focc,
      it_flight TYPE sbc400_t_sbc400focc.
PARAMETERS: pa_car TYPE s_carr_id.

* select all datasets from database table SFLIGHT
* corresponding to carrier 
SELECT carrid connid fldate seatsmax seatsocc
       FROM sflight
       INTO CORRESPONDING FIELDS OF wa_flight
       WHERE carrid = pa_car.

* calculate occupation of each flight
  wa_flight-percentage =
    100 * wa_flight-seatsocc / wa_flight-seatsmax.

* append single line into internal table
  INSERT wa_flight INTO TABLE it_flight.

* alternatively, you could use the APPEND statement,
* if you are using standard tables:
*
*    APPEND wa_flight TO it_flight.

ENDSELECT.

* sort internal table
SORT it_flight BY percentage.

* create list from sorted internal table
LOOP AT it_flight INTO wa_flight.
  WRITE: / wa_flight-carrid,
           wa_flight-connid,
           wa_flight-fldate,
           wa_flight-seatsocc,
           wa_flight-seatsmax,
           wa_flight-percentage,'%'.
ENDLOOP.

Solutions

	Unit:
Data Retrieval

Topic:
Authorization Checks


3
Model solution:

REPORT sapbc400dds_authority_check.

CONSTANTS actvt_display TYPE activ_auth VALUE '03'.

DATA wa_flight TYPE sbc400focc.
PARAMETERS pa_car TYPE s_carr_id.

* authority-check:
* Is user authorized to display data for carrier PA_CAR?
AUTHORITY-CHECK OBJECT 'S_CARRID'
    ID 'CARRID' FIELD pa_car
    ID 'ACTVT'  FIELD actvt_display.
CASE sy-subrc.

* user is authorized
  WHEN 0.
    SELECT carrid connid fldate seatsmax seatsocc
           FROM sflight
           INTO CORRESPONDING FIELDS OF wa_flight
           WHERE carrid = pa_car.

      wa_flight-percentage =
        100 * wa_flight-seatsocc / wa_flight-seatsmax.

      WRITE: / wa_flight-carrid COLOR COL_KEY,
               wa_flight-connid COLOR COL_KEY,
               wa_flight-fldate COLOR COL_KEY,
               wa_flight-seatsocc,
               wa_flight-seatsmax,
               wa_flight-percentage,'%'.
    ENDSELECT.

* user is not authorized or
* other error of authority-check
  WHEN OTHERS.
    WRITE: / 'Authority-Check Error'(001).
ENDCASE.

(REPORT sapbc400dds_authority_check_2 analog).

